Maple Fundamentals

Last revision: October 4, 2010

This tutorial is intended for beginning Calculus students. It is by no means a comprehensive introduction to maple. The purpose of this document is to help you get started on maple, become familiar with the basic maple commands, and solve elementary tasks with the help of maple. Maple was developed at the University of Waterloo in Ontario, Canada, hence the name and the leaf symbol.

Maple operates in the **worksheet mode** or in the document mode (default). Much of this tutorial is written in the **document mode**, as indicated by the red > symbol on the left. Occasionally we will switch to the document mode.

Basics

Maple is a computer algebra system. Unlike the standard graphing calculator it will attempt to give answers in **symbolic form**, rather than finding decimal approximations.

Example: Compute $\sqrt{8}$.

```
> sqrt(8);
```

$$2\sqrt{2} \quad \textbf{(1.1)}$$

```
> sqrt(8.0);
```

$$2.828427125 \quad \textbf{(1.2)}$$

In the first case maple simplified the result. When given a decimal input maple gave a decimal output in the second case.

Like any other programming language there are some **rules** which you need to follow. Here are some of them

1. Every command line must end with a semicolon. You may substitute a colon if you want to suppress the output.
2. Maple is case-sensitive, X and x mean different things.
3. A * for multiplication is required. This is different from graphing calculators were you can enter 2x. In maple 2*x is required.
4. You can use the # symbol for comments at the end of a command line (after the semicolon).
5. You may use % to refer to the last executed expression.

In 2-d input (default) you can get away without the semicolon and without the multiplication *. But be careful, even there the * is required in certain situations.

Illustrations:

```
> sqrt(5)*sqrt(20);  # multiply two roots
```

$$10 \quad \textbf{(1.3)}$$

```
> X-x;
```

$$X-x \quad \textbf{(1.4)}$$

```
> X-x:   # output suppressed
> %;     # use of %
```

$$X - x \quad (1.5)$$

```
> 2x;    # missing *
Error, missing operator or `;`
> 2*x;
```

$$2\,x \quad (1.6)$$

Here we show now work in the document mode. You can even use some symbols from the expression tab on the left.

sqrt(5)·sqrt(20);

$$10 \quad (1.7)$$

$\sqrt{5}\,\sqrt{20}$

$$10 \quad (1.8)$$

$2\,x$;

$$2\,x \quad (1.9)$$

$(x-1)\cdot(x-2)$; # *with multiplication* ·

$$(x-1)\,(x-2) \quad (1.10)$$

$(x-1)(x-2)$; # *without the* ·

$$x(x-2)-1 \quad (1.11)$$

There is a host of maple commands, built-in functions and the like. You can obtain information about a specific command using the ? key.

Example:

```
> ? sqrt
```

π and e

The constant π needs to be entered as Pi (upper case P), otherwise you just obtain a Greek letter without a numerical value attached to it. Verification:

```
> Pi;
```

$$\pi \quad (2.1)$$

```
> evalf(Pi);
```

$$3.141592654 \quad (2.2)$$

```
> pi;
```

$$\pi \quad (2.3)$$

```
> evalf(pi);
```

$$\pi \quad (2.4)$$

The Euler constant e is obtained as $f(1)$ when $f(x) = e^x$. Accordingly, it is entered as

```
> exp(1);
```

$$\mathrm{e} \quad (2.5)$$

A decimal representation can be found with

```
> exp(1.0);
```

$$2.718281828 \quad (2.6)$$

The common symbols tab on the left contains π and e and their use is straightforward.

π

$$\pi \quad (2.7)$$

$\xrightarrow{\text{at 10 digits}}$

$$3.141592654 \quad (2.8)$$

e

$$e \quad (2.9)$$

$\xrightarrow{\text{at 10 digits}}$

$$2.718281828$$

:= and restart

The **assignment operator** in maple is := (colon equals). You use it to assign a given expression to a new variable. For instance, suppose you want to assign the expression $x^2 - 3x + 3$ to the variable y. Here is how you do it:

```
> y := x^2 -4*x + 3;
```

$$y := x^2 - 4x + 3 \quad (3.1)$$

From here on, any time you reference the variable y, the quantity $x^2 - 4x + 3$ will b used in place of y.

Illustrations:

```
> y^2;
```

$$(x^2 - 4x + 3)^2 \quad (3.2)$$

```
> factor(y);
```

$$(x - 1)(x - 3) \quad (3.3)$$

```
> solve( y = 3,x);
```

$$0, 4 \quad (3.4)$$

In order to undo this definition you have two options. Either reset y to a string or use the restart command.

```
> y := 'y';
```

$$y := y \quad (3.5)$$

```
> y ;  # testing
```

$$y \quad (3.6)$$

```
> restart;
```

restart resets all variables, which have been defined in the course of the maple session.

Algebra

Factor, Simplify and Expand

Let's do some elementary algebra in maple. The commands **factor, expand and simplify** are extremely useful.

```
> factor( x^2 - 4*x + 3);
```

$$(x-1)\,(x-3) \tag{4.1.1}$$

```
> expand( (x-1)*(x-2)*(x-3)*(x-4) );
```

$$x^4 - 10\,x^3 + 35\,x^2 - 50\,x + 24 \tag{4.1.2}$$

```
> (x^(-1) +y^(-1))^(-1);
```

$$\frac{1}{\frac{1}{x} + \frac{1}{y}} \tag{4.1.3}$$

```
> simplify(%);
```

$$\frac{x\,y}{y+x} \tag{4.1.4}$$

```
>
```

Let's redo this in the document mode with 2-d input and right-clicks.

$x^2 - 4\,x + 3$

$$x^2 - 4\,x + 3 \tag{4.1.5}$$

$\overset{\text{factor}}{=}$

$$(x-1)\,(x-3) \tag{4.1.6}$$

$(x-1)\cdot(x-2)\cdot(x-3)\cdot(x-4)$

$$(x-1)\,(x-2)\,(x-3)\,(x-4) \tag{4.1.7}$$

$\overset{\text{expand}}{=}$

$$x^4 - 10\,x^3 + 35\,x^2 - 50\,x + 24;$$

$\left(x^{-1} + y^{-1}\right)^{-1}$

$$\frac{1}{\frac{1}{x} + \frac{1}{y}} \tag{4.1.8}$$

$\overset{\text{simplify}}{=}$

$$\frac{x\,y}{y+x} \tag{4.1.9}$$

Equations, Solve and Fsolve

A major part of algebra deals with **solving equations**. In maple we have the **solve** command to accomplish this task. The syntax is solve(equation, unknown), that is, you need to communicate the equation which needs solving and the variable which you are trying to find. The **evalf** command comes in handy when you look for numerical values rather than symbolic answers.

Example: Solve the equation $x^3 + x + 1 = 3\,x^2$.

Note that for equations we use the simple = sign, and not :=. Let's first use **solve** in this problem.

```
> solve( x^3 + x + 1 = 3*x^2,x);
```

$$1,\ 1+\sqrt{2},\ 1-\sqrt{2} \tag{4.2.1}$$

maple gives symbolic answers. We can get decimal conversions with **evalf**

```
> evalf(%);
```

$$1.,\ 2.414213562,\ -0.414213562 \tag{4.2.2}$$

Numerical answers can also be obtained directly with the **fsolve** command

```
> fsolve( x^3 + x + 1 = 3*x^2,x);
```

$$-0.4142135624,\ 1.,\ 2.414213562 \tag{4.2.3}$$

Example: Solve the equation $\sin(2x) = \sin(x)$.
We know that this equation has infinitely many solutions, maple selects a few. First we define the equation as Eqn, and then we solve it with **solve**.

```
> Eqn := sin(x) = sin(2*x);
```

$$Eqn := \sin(x) = \sin(2x) \tag{4.2.4}$$

```
> solve(Eqn, x);
```

$$\frac{1}{3}\pi,\ -\frac{1}{3}\pi,\ 0,\ \pi \tag{4.2.5}$$

```
> evalf(%);
```

$$1.047197551,\ -1.047197551,\ 0.,\ 3.141592654 \tag{4.2.6}$$

fsolve for the same problem results in

```
> fsolve( Eqn , x);
```

$$0. \tag{4.2.7}$$

fsolve employs a numerical method and it singles out just one solution. But fsolve also lets you select a search interval or an initial point. If we want solutions in the interval [1,3] we enter

```
> fsolve(Eqn,x=1..3);
```

$$1.047197551 \tag{4.2.8}$$

A solution near $x = 5$ can be found by

```
> fsolve(Eqn,x=5);
```

$$5.235987756 \tag{4.2.9}$$

Example: Find the **inverse function** of $f(x) = \frac{x-1}{x-2}$.

We know from class that we need to solve y=f(x) for y. Let's do it.

```
> y = (x-1)/(x-2);
  solve(%,x);
```

$$y = \frac{x-1}{x-2}$$

$$\frac{2y-1}{y-1} \tag{4.2.10}$$

Thus, the inverse function is $g(x) = \frac{2x-1}{x-1}$.

Finally, we repeat the computation using **clickable maple**
$\sin(x) = \sin(2x)$;

$$\sin(x) = \sin(2\,x) \qquad (4.2.11)$$

$\xrightarrow{\text{solve}}$

$$\left\{x = \frac{1}{3}\,\pi\right\},\ \left\{x = -\frac{1}{3}\,\pi\right\},\ \{x = 0\},\ \{x = \pi\}$$

$\sin(x) = \sin(2\,x);$

$$\sin(x) = \sin(2\,x) \qquad (4.2.12)$$

$\xrightarrow{\text{solve}}$

$$5.235987756 \qquad (4.2.13)$$

$y = \frac{x-1}{x-2}$

$$y = \frac{x-1}{x-2} \qquad (4.2.14)$$

$\xrightarrow{\text{solve for x}}$

$$\left[\left[x = \frac{2\,y-1}{y-1}\right]\right]$$

Functions

```
> restart;
```

It is essential that you understand the difference between functions, expressions and equations. Let's do this by example.

- $x^2 - 4\,x + 3$ is an expression. It can be factored, simplified or whatever else you might want to do with it. But it cannot be solved, because it is not an equation.
- $y = x^2 - 4\,x + 3$ is an equation (note the = symbol). Equations can be solved, or they can be graphed (all points in the xy-plane whose coordinates satisfy the equation).
- $f(x) = x^2 - 4\,x + 3$ denotes a function. The input is x, and $x^2 - 4\,x + 3$ is the formula for the output. Maple uses the -> syntax in function definitions.

Example: Define the function $f(x) = x^2 - 4\,x + 3$ and compute $f(4), f(-x), f(x+3)$ and $\frac{f(x+h) - f(x)}{h}$.

```
> f := x -> x^2-4*x+3;
```

$$f := x \rightarrow x^2 - 4\,x + 3 \qquad (5.1)$$

```
> f(4);
```

$$3 \qquad (5.2)$$

```
> f(-x);
```

$$x^2 + 4\,x + 3 \qquad (5.3)$$

```
> f(x+3);
```

$$(x+3)^2 - 4\,x - 9 \qquad (5.4)$$

```
> simplify(%);
```

$$x^2 + 2\,x \qquad (5.5)$$

```
> (f(x+h)-f(x))/h;
```

$$\frac{(x+h)^2 - 4\,h - x^2}{h} \tag{5.6}$$

```
> simplify(%);
```

$$2\,x + h - 4 \tag{5.7}$$

Example: The same example in clickable maple.

$f := x \rightarrow x^2 - 4\,x + 3$

$$x \rightarrow x^2 - 4\,x + 3 \tag{5.8}$$

$f(x);$

$$x^2 - 4\,x + 3 \tag{5.9}$$

$\xrightarrow{\text{evaluate at point}}$

$$3$$

$f(-x);$

$$x^2 + 4\,x + 3 \tag{5.10}$$

$f(x+3);$

$$(x+3)^2 - 4\,x - 9 \tag{5.11}$$

$\overset{\text{simplify}}{=}$

$$x^2 + 2\,x \tag{5.12}$$

$\frac{f(x+h) - f(x)}{h}$

$$\frac{(x+h)^2 - 4\,h - x^2}{h} \tag{5.13}$$

$\overset{\text{simplify}}{=}$

$$2\,x + h - 4 \tag{5.14}$$

Here is a list of common functions, most of which can be found on the expressions tab on the left.

```
> sqrt(x);      # square root
  root[5](x);   # 5th order root
  abs(x);       # absolute value
  exp(x);       # exponential function
  ln(x);        # natural logarithm
  sin(x);       # sine function
  cos(x);       # cosine function
  tan(x);       # tangent function
  sec(x);       # secant function
  arcsin(x);    # inverse sine function
  arctan(x);    # inverse tangent function
```

$$\sqrt{x}$$

$$x^{1/5}$$

$$|x|$$
$$e^x$$
$$\ln(x)$$
$$\sin(x)$$
$$\cos(x)$$
$$\tan(x)$$
$$\sec(x)$$
$$\arcsin(x)$$
$$\arctan(x) \quad \textbf{(5.15)}$$

```
>
```

More on Functions

Even and Odd Functions

Example: Test whether the function $f(x) = \sin(x \cdot \cos(x))$ is even or odd.
First we define f

```
> f := x -> sin(x*cos(x));
```

$$f := x \to \sin(x\cos(x)) \quad \textbf{(6.1.1)}$$

Now we compute f(-x) and compare it to f(x).

```
> f(-x);
```

$$-\sin(x\cos(x)) \quad \textbf{(6.1.2)}$$

```
> f(x);
```

$$\sin(x\cos(x)) \quad \textbf{(6.1.3)}$$

Since f(x) and f(-x) are negatives of each other, the function f is an odd function.

Example: For any function f(x) the function g(x) = f(|x|) will be an even function.

```
> restart;
> g := x -> f(abs(x));
```

$$g := x \to f(|x|) \quad \textbf{(6.1.4)}$$

```
> g(-x);
```

$$f(|x|) \quad \textbf{(6.1.5)}$$

We see that g(-x) = g(x) and thus g is an even function. Let's illustrate this for the case where $f(x) = \ln(x+5)$.

```
> f := x -> ln(x+5);
```

$$f := x \to \ln(x+5) \quad \textbf{(6.1.6)}$$

```
> g(x);
```

$$\ln(|x|+5) \quad \textbf{(6.1.7)}$$

```
> g(-x);
```

$$\ln(|x|+5) \quad \textbf{(6.1.8)}$$

```
> g(x) - g(-x);
```

(6.1.9)

$$0 \quad \textbf{(6.1.9)}$$

Composition of Functions

The composition of functions is not a problem in maple.

Example: Let $f(x) = \sqrt{x+1}$ and let $g(x) = x^2 - 4x + 5$. Then compute and simplify f(g(x)) and g(f(x)).

First we define the functions

```
> f := x -> sqrt(x+1);
  g := x -> x^2 -4*x+5;
```

$$f := x \to \sqrt{x+1}$$

$$g := x \to x^2 - 4x + 5 \quad \textbf{(6.2.1)}$$

Now compute the compositions.

```
> f(g(x));
```

$$\sqrt{x^2 - 4x + 6} \quad \textbf{(6.2.2)}$$

```
> g(f(x));
```

$$x + 6 - 4\sqrt{x+1} \quad \textbf{(6.2.3)}$$

Inverse Functions

In the calculation of inverse functions we do exactly the steps from class, namely, solve y=f(x) for x, and then interchange x and y (or interchange x and y first, if you prefer).

Example: Find the inverse function of $f(x) = \ln(x-5)$.

```
> f := x -> ln(x-5);    # define f(x)
```

$$f := x \to \ln(x-5) \quad \textbf{(6.3.1)}$$

```
> solve( y = f(x) , x);    # solve for x
```

$$e^y + 5 \quad \textbf{(6.3.2)}$$

```
> g := x -> exp(x) + 5;    # interchange x and y and define
  the inverse
```

$$g := x \to e^x + 5 \quad \textbf{(6.3.3)}$$

Now let's test the result.

```
> g(f(x));
```

$$x \quad \textbf{(6.3.4)}$$

```
> f(g(x));
```

$$\ln(e^x) \quad \textbf{(6.3.5)}$$

```
> simplify(%);
```

$$\ln(e^x) \quad \textbf{(6.3.6)}$$

It appears that maple has trouble with the identity $x = \ln(e^x)$. Otherwise we get the expected results.

Example: Find the inverse function of $f(x) = \frac{\sin(ax)}{b}$ for arbitrary constants a and b.

```
> f := x -> sin(a*x)/b;
```

$$f := x \to \frac{\sin(a\,x)}{b} \quad \textbf{(6.3.7)}$$

```
> g := x -> solve( x=f(y),y);  #getting fancy here and do it
  all in one step
```

$$g := x \to solve(x = f(y), y) \quad \textbf{(6.3.8)}$$

```
> g(x);    # Voila!
```

$$\frac{\arcsin(x\,b)}{a} \quad \textbf{(6.3.9)}$$

Let's test the result.

```
> g(f(x));
```

$$x \quad \textbf{(6.3.10)}$$

```
> f(g(x));
```

$$x \quad \textbf{(6.3.11)}$$

It works, provided that neither a nor b are zero.

Graphing

```
> restart;
```

The command for graphing has the structure plot(expression , range, options). It you don't give a range, maple will use the interval where $10 \le x \le 10$. The range for y will always be selected so that the graph fits, which can lead to awkward pictures if the graph has a horizontal asymptote.

Example: Plot the function $f(x) = x^2 - 4\,x + 3$ on the interval [-2,5].

```
> f := x -> x^2 -4*x+3;
```

$$f := x \to x^2 - 4\,x + 3 \quad \textbf{(7.1)}$$

```
> plot(f);  # not correct, did not specify the interval
```




```
> plot(f,-2..5);
```




```
> plot(f(x), x= -2...5);
```



Note that both plot(f,-2..5) and plot(f(x), x=-2..5) give the same graph. Rule of thumb: either you mention x with f(x) and the range, or you don't include x at all. These commands do not mix.

The graphs usually come out fairly large in maple. It is recommended to manually resize them so that you will save on paper. Use the print previewer (fifth icon from the left) for your decision.

Now let's add some bells and whistles to the graph.

view(x-range, y-range) sets the viewing area just like on a graphing calculator.

thickness controls the thickness of the curve
color sets the color (duh)
title let's you include a title
numpoints: a high number will refine the graph and take out unwanted corners
discont=true will allow for jumps in the graph.
scaling=constrained will use the same length units on the x- and the y-axis (square option on a graphing calculator).

Much of this can be done by selecting the **Plot Builder** option.

Example: Graph the function $f(x) = \dfrac{x^2+1}{x-1}$.

```
> f := x -> (x^2+1)/(x-1);
```

$$f := x \to \frac{x^2+1}{x-1} \tag{7.2}$$

```
> plot(f,-2..5,view=[-2..5,-3..10], discont=true,color=blue,
  thickness=2,title=`what a graph...`,numpoints=5,scaling=
  constrained);
```



The noticeable corners are caused by selecting a very small value in numpoints (bad idea).

If you want to display **several functions in a common picture** you need to enclose the functions in a common bracket. You can select different colors or a different thickness for each curve by using brackets.

Example: Display $f(x) = \cos\left(\dfrac{\pi x}{2}\right)$ in red along with a shift of this curve. The shift should be up by 2 units and to the left by one unit. Display the shifted curve in blue.

```
> f:= x -> cos(Pi*x/2);
```

$$f := x \to \cos\left(\frac{1}{2}\pi x\right) \tag{7.3}$$

```
> plot([f(x),f(x-1)+2],x=-2..8,color=[red,blue],scaling=
  constrained);
```



Example: Display the function $f(x) = \frac{1}{2} + \ln(x)$ along with its inverse in a common graph.

```
> f := x -> 1/2 + ln(x);  # define f
```

$$f := x \to \frac{1}{2} + \ln(x) \quad (7.4)$$

```
> solve( y=f(x), x);   # inverse function
```

$$e^{y - \frac{1}{2}} \quad (7.5)$$

```
> g := x -> exp(x-1/2); # manually define the inverse function
```

$$g := x \to e^{x - \frac{1}{2}} \quad (7.6)$$

```
> plot([f(x),g(x)],x=-3..3, view=[-3..3,-3..3],color=[red,blue]
  );
```



Let's include the diagonal to make the symmetry more evident.

```
> plot([f(x),g(x),x],x=-3..3, view=[-3..3,-3..3],color=[red,
  blue,black],thickness=[2,2,1]);
```


3
2
1
-3
-2
1
2
3
-1
x
-2
-3

>